


1. แนะนำ

1.1. ภาพประกอบบอร์ด ESP32 (30 ขา)



1.2. ตารางสรุปขาที่ใช้งานบนโมดูล ESP32 (30 ขา)

หมายเลขขา	ขา GPIO	INPUT	OUTPUT	หมายเหตุ
1	-	-	-	VIN ไฟบวก 5 โวลต์
2	-	-	-	GND
3	GPIO13	ได้	ได้	ใช้งานทั่วไปได้
4	GPIO12	ได้	ระวัง	ห้าม HIGH ตอนบูท
5	GPIO14	ได้	ได้	ใช้งานทั่วไปได้
6	GPIO27	ได้	ได้	ใช้งานทั่วไปได้
7	GPIO26	ได้	ได้	DAC2
8	GPIO25	ได้	ได้	DAC1
9	GPIO33	ได้	ได้	ADC1
10	GPIO32	ได้	ได้	ADC1
11	GPIO35	ได้	ไม่ได้	Input only
12	GPIO34	ได้	ไม่ได้	Input only
13	GPIO39	ได้	ไม่ได้	ADC1 input only
14	GPIO36	ได้	ไม่ได้	ADC1 input only
15	EN	-	-	Enable
16	-	-	-	3.3V
17	-	-	-	GND

18	GPIO15	ได้	ระวัง	ต้อง LOW ตอนบูท
19	GPIO2	ได้	ได้	ต้อง HIGH ตอนบูท แต่ใช้งานทั่วไปได้
20	GPIO4	ได้	ได้	ใช้งานทั่วไปได้
21	GPIO16	ได้	ได้	ใช้งานทั่วไปได้
22	GPIO17	ได้	ได้	ใช้งานทั่วไปได้
23	GPIO5	ได้	ได้	ใช้งานทั่วไปได้
24	GPIO18	ได้	ได้	ใช้งานทั่วไปได้
25	GPIO19	ได้	ได้	ใช้งานทั่วไปได้
26	GPIO21	ได้	ได้	I2C SDA
27	GPIO3 (RX0)	ได้	ได้ (ไม่แนะนำ)	มีการรับ Serial ตอนบูท
28	GPIO1 (TX0)	ได้	ได้	ส่ง Serial ตอนบูท มีข้อความรบกวน
29	GPIO22	ได้	ได้	I2C SCL
30	GPIO23	ได้	ได้	ใช้งานทั่วไปได้


1.3. ข้อควรระวังในการใช้งานขา GPIO

ขา GPIO	ข้อควรระวัง
GPIO2	ต้อง HIGH ตอนบูท แต่ใช้งานทั่วไปได้
GPIO12	ห้าม HIGH ตอนบูท
GPIO15	ต้อง LOW ตอนบูท
GPIO35	Input only
GPIO34	Input only
GPIO39	ADC1 input only
GPIO36	ADC1 input only
GPIO3 (RX0)	ใช้ได้ ไม่แนะนำ
GPIO1 (TX0)	ใช้ได้ ไม่แนะนำ


2. เตรียมเครื่องมือและติดตั้งโปรแกรม

2.1. ดาวน์โหลดโปรแกรม Arduino IDE


เข้าที่เว็บไซต์ Arduino : <https://www.arduino.cc/en/software/>



2.2. ติดตั้งโปรแกรม




2.3. เปิดโปรแกรมขึ้นมาแล้ว ให้ไปที่เมนู File -> Preferences เพื่อติดตั้งบอร์ด ESP32




2.4. เพิ่ม https://dl.espressif.com/dl/package_esp32_index.json

ลงในช่อง Additional Boards Manager




2.5. คลิกไปที่เมนู Tools -> Board -> Board Manager



2.6. พิมพ์คำว่า ESP32 เพื่อค้นหาลงในช่อง และเริ่มต้นติดตั้งดังภาพ




2.7. เมื่อติดตั้งเสร็จสิ้น ในหน้าต่าง Board ก็จะมีปรากฏประเภทของบอร์ด ESP32 ขึ้นมา ให้เลือกใช้งานได้แล้ว



2.8. เริ่มต้นใช้งานบอร์ด ESP32 โดยเชื่อมต่อสาย USB เข้ากับตัวบอร์ด



2.9. เลือก Port ที่เชื่อมต่อบอร์ด ESP32



เสร็จสิ้นขั้นตอนการติดตั้งโปรแกรม และพร้อมใช้งาน


3. ความรู้การเขียนโปรแกรมมิ่ง และคำสั่งที่ใช้งาน

3.1. ชนิดตัวแปร

ชนิด	ข้อมูลที่เก็บ	ขนาด	ช่วงค่า
boolean	true/false	1 byte	-
char	ตัวอักษร 1 ตัว	1 byte	-128 ถึง 127
byte	จำนวนเต็มไม่ติดลบ	1 byte	0 ถึง 255
int	จำนวนเต็ม	4 byte	-2.1B ถึง 2.1B
unsigned int	จำนวนเต็มบวก	4 byte	0 ถึง 4.29B
long	จำนวนเต็ม	4 byte	-2.1B ถึง 2.1B
unsigned long	จำนวนเต็มบวก	4 byte	0 ถึง 4.29B
float	เลขทศนิยม	4 byte	ประมาณ 7 หลัก
double	ทศนิยมละเอียดสูง	8 byte	ประมาณ 15 หลัก
void	ไม่มีค่า	-	-
String	ข้อความ	dynamic	-

3.2. การประกาศตัวแปร

3.2.1. รูปแบบการประกาศตัวแปร



3.2.2. กฎในการตั้งชื่อตัวแปร

- ☐ ขึ้นต้นด้วย ตัวอักษร หรือ _ (underscore) ได้
- ☐ ห้ามขึ้นต้นด้วยตัวเลข
- ☐ ใช้ตัวอักษร a-z, A-Z, 0-9 และ _ ได้
- ☐ ห้ามใช้ชื่อที่เป็น Keyword ของ C/C++ เช่น if, while, for, return, void, function
- ☐ ตัวพิมพ์เล็ก-ใหญ่ ถือว่าเป็นคนละตัว เช่น temp ☐ Temp

3.2.3. การประกาศตัวแปรแบบกำหนดค่าในบรรทัดเดียว

ตัวอย่าง

```
int led = 2;
```

```
boolean status = false;
```

3.2.4. การประกาศตัวแปรแบบประกาศก่อน กำหนดค่าทีหลัง

ตัวอย่าง

```
int led;
boolean status;
led = 2;
status = false;
```

3.2.5. การประกาศตัวแปรหลายตัวในบรรทัดเดียว

ตัวอย่าง

```
int a, b, c;
```

```
int a = 10, b = 20, c = 30;
```

3.2.6. การประกาศตัวแปรแบบ Global มองเห็นได้ทั้งโปรแกรม

ตัวอย่าง

```
int counter = 0;    // Global
void setup() {
  Serial.begin(115200);
}
void loop() {
  counter++;
  Serial.println(counter);
}
```

ตัวแปร counter เข้าถึงได้ในทุกฟังก์ชันเหมือนกับค่าที่ต้องใช้ร่วมกันหลายส่วนของโปรแกรม

3.2.7. ตัวแปรแบบ Local มองเห็นเฉพาะในฟังก์ชัน

ตัวอย่าง

```
void setup() {
  Serial.begin(115200);
}
void loop() {
  int counter = 10;    // Local
  Serial.println(counter);
}
```

ตัวแปร counter ใช้ได้เฉพาะในฟังก์ชัน loop เท่านั้น นอกฟังก์ชันจะใช้ไม่ได้

3.2.8. ตัวแปรแบบ const (ค่าคงที่ เปลี่ยนค่าไม่ได้)

ใช้แทนค่าที่ไม่ควรถูกเปลี่ยน เช่น หมายเลขขา หรือค่าคงที่ทางคณิตศาสตร์

ตัวอย่าง

```
const int LED_PIN = 2;
const float PI = 3.14159;
```

3.2.9. ตัวอย่างการประกาศตัวแปร

ตัวอย่าง

```
boolean status = false;
int count;
int pin = 4;
float temperature = 25.5;
float humidity;
char letter = 'A';
String message;
```


3.3. คำสั่ง if else เช็คเงื่อนไข

ใช้สำหรับตรวจสอบว่าเงื่อนไขเป็น จริง (true) หรือ เท็จ (false) เพื่อตัดสินใจทำหรือไม่ทำ

รูปแบบ

```
if (เงื่อนไข) {
    // คำสั่งเมื่อเงื่อนไขเป็นจริง
} else {
    // คำสั่งเมื่อเงื่อนไขเป็นเท็จ
}
```

ตัวอย่าง

```
int a = 4;
int b = 5;
void setup() {
    Serial.begin(115200);
    if( a > b ) {
        Serial.println("a more than b");
    } else {
        Serial.println("a less than b");
    }
}
void loop() {
}
```

3.4. คำสั่งวนลูป while, for

3.4.1. คำสั่งวนลูป while

ใช้เมื่อ “ยังไม่รู้จำนวนครั้งที่แน่นอน” แต่ต้องการให้ทำงานกว่าเงื่อนไขจะเป็นเท็จ

รูปแบบ

```
while (เงื่อนไข) {
    // คำสั่งทำซ้ำ
}
```

ตัวอย่าง รอนจนกว่าปุ่มจะถูกกด

```
while (digitalRead(5) == LOW) {
    Serial.println("รอนการกดปุ่ม...");
    delay(200);
}
Serial.println("กดปุ่มแล้ว!");
```

3.4.2. คำสั่งวนลูป for

ใช้เมื่อ “รู้จำนวนครั้งที่แน่นอน” แต่ต้องการให้ทำงานกว่าเงื่อนไขจะเป็นเท็จ
รูปแบบ

```
for (ตัวแปรเริ่มต้น; เงื่อนไข; การเปลี่ยนค่า) {
    // คำสั่งที่ต้องการทำซ้ำ
}
```

ตัวอย่าง พิมพ์ตัวเลข 1 ถึง 5

```
for (int i = 1; i <= 5; i++) {
    Serial.println(i);
}
```

3.5. ตารางสรุปคำสั่งที่ใช้งานบ่อย

คำสั่ง	คำอธิบาย	ตัวอย่าง
Serial.begin(baudrate);	ใช้การสื่อสารระหว่าง ESP32 □ คอมพิวเตอร์ สามารถกำหนด baudrate ความเร็วในการส่งข้อมูล โดยทั่วไป 2400 , 4800 , 9600 , 14400 , 38400 , 57600 , 115200 , 230400 , 460800 , 921600	Serial.begin(115200);
Serial.print(data);	สั่งให้พิมพ์ข้อมูล data	Serial.print("Hello Word");
Serial.println(data);	สั่งให้พิมพ์ข้อมูล data พร้อมกับขึ้นบรรทัดใหม่	Serial.println("Hello Word");
pinMode(pin, mode)	ใช้กำหนดโหมดของขา pin ให้เป็น INPUT รับค่าจากภายนอก ต้องต่อตัวต้านทานเพิ่ม (เช่น ปุ่มกด เซนเซอร์) OUTPUT ส่งค่าออกไปภายนอก (เช่น LED, Relay, Buzzer) INPUT_PULLUP รับค่าจากภายนอก ใช้ตัวต้านทาน Pull-Up ภายใน	pinMode(2, OUTPUT);
digitalWrite(pin, value)	ใช้เขียนค่าดิจิทัลออกไปที่ขา pin HIGH = 1 (ไฟติด) LOW = 0 (ไฟดับ)	digitalWrite(2, HIGH);
digitalRead(pin);	อ่านค่าข้อมูลที่ขา pin เป็น HIGH หรือ LOW	int button=digitalRead(2);

analogRead(pin)	ใช้สำหรับอ่านค่าแรงดันไฟฟ้าที่ขา pin อยู่ในช่วง 0–4095 ADC (Analog-to-Digital Converter)	int button=analogRead(4);
analogWrite(pin, value)	ใช้เขียนค่า analog ไปที่ขา pin ช่วงตั้งแต่ 0 ถึง 255 โดยที่ 0 คือไฟดับ และ 255 ไฟสว่างมาก	digitalWrite(2, 120);
delay(ms)	ใช้หยุดโปรแกรมตามเวลาที่กำหนด (มิลลิวินาที) 1000 ms = 1 วินาที ระหว่าง delay() โปรแกรมจะไม่ทำอย่างอื่นเลย	delay(500); // หยุดโปรแกรมครึ่งวินาที
map(value, fromLow, fromHigh, toLow, toHigh)	แปลงค่าจากช่วงหนึ่งไปสู่อีกช่วงหนึ่ง เช่น เซนเซอร์ให้ค่า 0–1023 ต้องแปลงให้อยู่ในช่วง 0–100 (%)	int value_new = map(value_old, 0, 1023, 0, 100);

4. การใช้งาน Serial Monitor พิมพ์ Hello World

ใช้ทำอะไร ?

- ☐ Serial คือการสื่อสารระหว่าง ESP32 ☐ คอมพิวเตอร์
- ☐ ใช้ debug โปรแกรม แสดงค่าต่าง ๆ

ตัวอย่างโค้ดที่ 1 การแสดงข้อความ Hello World ผ่าน Serial Monitor

```
void setup() {
  Serial.begin(115200);
}
void loop() {
  Serial.println("Hello World");
  delay(1000);
}
```

ตัวอย่างโค้ดที่ 2 การรับค่าข้อความ และแสดงผลออกมา ผ่าน Serial Monitor

```
void setup() {
  Serial.begin(115200);
}
void loop() {
  if (Serial.available()) { // ตรวจสอบว่ามีข้อมูลเข้ามาหรือไม่
    String data = Serial.readString(); // อ่านค่าเข้ามา
    data.trim();
    Serial.print("Received: ");
    Serial.println(data); // แสดงผลค่าที่ได้รับ
  }
}
```

5. การใช้งาน Digital Output


ใช้เขียนค่าดิจิทัลออกไปที่ขา pin โดยที่ HIGH = 1 (ไฟติด) และ LOW = 0 (ไฟดับ)

ใช้ทำอะไร ?

- ☐ ควบคุม LED
- ☐ เปิด-ปิด Relay หรือไฟบ้าน
- ☐ ส่งสัญญาณ ON/OFF


ตัวอย่างโค้ดที่ 3 ไฟกระพริบเปิด-ปิด 1 ดวง

```
int led = 2;
void setup() {
  pinMode(led, OUTPUT);
}
void loop() {
  digitalWrite(led, HIGH);
  delay(500);
  digitalWrite(led, LOW);
  delay(500);
}
```



ตัวอย่างโค้ดที่ 4 ไฟกระพริบสลับเปิด-ปิด 2 ดวง

```
int led1 = 2;
int led2 = 4;
void setup() {
  pinMode(led1, OUTPUT);
  pinMode(led2, OUTPUT);
}
void loop() {
  digitalWrite(led1, HIGH);
  digitalWrite(led2, LOW);
  delay(500);
  digitalWrite(led1, LOW);
  digitalWrite(led2, HIGH);
  delay(500);
}
```



6. การใช้งาน Digital Input


อ่านค่าข้อมูลที่ขา pin เป็น HIGH หรือ LOW

ใช้ทำอะไร ?

- ☐ ใช้ตรวจจับ ON/OFF
- ☐ ใช้ร่วมกับ Pull-up resistor


ตัวอย่างโค้ดที่ 5 การใช้ปุ่มกด โดยใช้ตัวต้านทาน Pull-Up ภายใน

```
int button = 13;
void setup(){
  pinMode(button, INPUT_PULLUP);
  Serial.begin(115200);
}
void loop(){
  if(digitalRead(button) == LOW){
    Serial.println("Button Pressed");
  }
}
```



ตัวอย่างโค้ดที่ 6 การใช้ปุ่มกด โดยใช้ตัวต้านทาน Pull-Up ภายนอก

```
int button = 13;
void setup(){
  pinMode(button, INPUT);
  Serial.begin(115200);
}
void loop(){
  if(digitalRead(button) == LOW){
    Serial.println("Button Pressed");
  }
}
```




7. การใช้งาน Analog Input

ใช้ทำอะไร ?

- ☐ อ่านค่าจากตัวต้านทานปรับค่าได้
- ☐ อ่านค่าจากเซนเซอร์ต่าง ๆ

ตัวอย่างโค้ดที่ 7 การอ่านค่าตัวต้านทานปรับค่าได้

```
int r = 4;
void setup() {
  pinMode(r, INPUT);
  Serial.begin(115200);
}
void loop() {
  int value = analogRead(r);
  Serial.println(value);
  delay(100);
}
```



8. การใช้งาน Analog Output


ปกติ ESP32 ไม่มี analogWrite จริง แต่เป็น PWM (LEDC)

ใช้ทำอะไร ?

- ☐ ควบคุมความเร็วมอเตอร์ การเพิ่มลดความสว่าง

ตัวอย่างโค้ดที่ 8 ควบคุมความสว่างหลอดไฟ LED

```
int r = 4;
int led = 2;
void setup() {
  pinMode(r, INPUT_PULLUP);
  pinMode(led, OUTPUT);
  Serial.begin(115200);
}
void loop() {
  int value = analogRead(r);
  int value_map = map(value, 0, 4095, 0, 255);
  Serial.print(value);
  Serial.print("\t");
  Serial.println(value_map);
  analogWrite(led, value_map);
  delay(100);
}
```



9. การใช้งาน WiFi

ใช้ทำอะไร ?

- ☐ แสดงผลค่าต่าง ๆ บนอินเทอร์เน็ต มือถือ หรือสั่งงานผ่านอินเทอร์เน็ต

9.1. ตัวอย่างโค้ดที่ 9 การเชื่อมต่อ WiFi

```
#include <WiFi.h>
const char* ssid = "iPhone 16";
const char* pass = "1234567890";
void setup() {
  Serial.begin(115200);
  WiFi.begin(ssid, pass);
  while(WiFi.status() != WL_CONNECTED) {
    Serial.print(".");
    delay(500);
  }
  Serial.println("WiFi Connected.");
  Serial.print("IP = ");
  Serial.println(WiFi.localIP());
}
void loop() {}
```

9.2. ตัวอย่างโค้ดที่ 10 เปิดโหมด Access Point เพื่อทำ Web Server

Library ที่ใช้: 1.ESPAsyncWebServer(by lacamera)v3.1.0
2.AsyncTCP (by dvarrel)v1.1.4

- ☐ ไฟล์โปรแกรมหลัก (esp32.ino)

```
#include <WiFi.h>
#include <ESPAsyncWebServer.h>
char* ssid = "My ESP32";
char* password = "12345678";
AsyncWebServer server(80);
#include "index.h"
void setup() {
  Serial.begin(9600);
  WiFi.softAP(ssid, password);
  delay(100);
  server.on("/", HTTP_GET, [] (AsyncWebServerRequest* request) {
    String html = index_page;
    request->send(200, "text/html", html);
  });
  server.begin();
}
void loop() {
}
```

□ ไฟล์หน้าเว็บโค้ด HTML (index.h)

```
const char index_page[] PROGMEM = R"=====(
<!DOCTYPE html>
<html>
<head>
  <meta charset="utf-8">
  <meta name="viewport" content="width=device-width, initial-
scale=1">
  <title>ESP32 Web Server</title>
</head>
<body>
  <h1>Hello World</h1>
  <h4>ยินดีต้อนรับสู่ ESP32 Web Server</h4>
</body>
</html>
)=====";
```

อัปโหลดโค้ดโปรแกรม และใช้มือถือค้นหาสัญญาณ WiFi ชื่อ “My ESP32” และเชื่อมต่อด้วยรหัสผ่าน 12345678 เมื่อเชื่อมต่อสำเร็จ เข้าไปเปิด Browser พิมพ์ URL : 192.168.4.1 จะปรากฏหน้าเว็บที่เราเขียนขึ้นมา

9.3. ตัวอย่างโค้ดที่ 11 การเชื่อมต่อ WiFi ไปบันทึกลงบน Google Sheet

□ ส่วนของ ESP32

```
#include <WiFi.h>
const char* ssid = "YOUR_SSID";
const char* pass = "YOUR_PASSWORD";
#include <HTTPClient.h>
void setup() {
  Serial.begin(115200);
  WiFi.begin(ssid, pass);
  while(WiFi.status() != WL_CONNECTED){
    Serial.print(".");
    delay(500);
  }
  Serial.println("WiFi Connected.");
  Serial.print("IP = ");
  Serial.println(WiFi.localIP());
  delay(1000);
  HTTPClient http;
  String url = "https://script.google.com/macros/s/AKfycbz-
PJ4Has5tsriMicKAMR_chvnq4Mt3l0SSyvbze00OfeZbV-
R0fsUxr4fz7_AgwTu9/exec?data=12.6";
  http.begin(url);
  int httpCode = http.GET();
  if(httpCode == 200) {
    String response = http.getString();
    Serial.print("Send Success : ");
    Serial.println(response);
  } else {
    Serial.println("Error code : " + httpCode);
  }
  http.end();
}
void loop() { }
```


□ ส่วนของ Google Sheet (App Script)

```
function doGet(e) {
  var data = e?.parameter["data"] ?? "";
  data = data.trim();
  if( data=="") {
    return ContentService.createTextOutput("Invalid Data");
  }
  var sheet_id = "1hYCBALPCUsUyvWE1wbnFVSwy8sghm785BhuyYHOnCb0";
  var sheet = SpreadsheetApp.openById(sheet_id).getSheetByName("data");
  var lib = 'https://cdnjs.cloudflare.com/ajax/libs/moment.js/2.18.1/moment.min.js';
  eval(UrlFetchApp.fetch(lib).getContentText());
  var date = moment().format("YYYY-MM-DD HH:mm:ss");
  sheet.appendRow([date, data]);
  return ContentService.createTextOutput("Save Success.");
}
```

10. การใช้งานหลอดไฟ LED 3 สี


ตัวอย่างโค้ดที่ 12 ไฟกระพริบสลับสีแดง สีเขียว สีน้ำเงิน

```
int ledR = 2;
int ledG = 4;
int ledB = 16;

void setup() {
  pinMode(ledR, OUTPUT);
  pinMode(ledG, OUTPUT);
  pinMode(ledB, OUTPUT);
}

void loop() {
  setColor(255, 0, 0); // แดง
  delay(500);
  setColor(0, 255, 0); // เขียว
  delay(500);
  setColor(0, 0, 255); // น้ำเงิน
  delay(500);
}

void setColor(int r, int g, int b) {
  analogWrite(ledR, r);
  analogWrite(ledG, g);
  analogWrite(ledB, b);
}
```




11. การใช้งานเสียง Buzzer

ตัวอย่างโค้ดที่ 13 การเปิด-ปิดเสียงแบบง่าย Active Buzzer (Low Trigger)

```
int pin = 2;
void setup() {
  pinMode(pin, OUTPUT);
}

void loop() {
  digitalWrite(pin, HIGH);
  delay(500);
  digitalWrite(pin, LOW);
  delay(500);
}
```




12. การใช้งานจอ LCD

ตัวอย่างโค้ดที่ 14 การแสดงข้อความบนจอ LCD ขนาด 16x2

Library ที่ใช้: LiquidCrystal_I2C (by Martin...) v2.0.0

```
#include <LiquidCrystal_I2C.h>
LiquidCrystal_I2C lcd(0x27, 16, 2);
void setup() {
  lcd.init();
  lcd.backlight();
  lcd.setCursor(0, 0);    // ไปที่ตัวอักษรที่ 0 แถว
  ที่ 1
  lcd.print("Welcome To");
  lcd.setCursor(6, 1);    // ไปที่ตัวอักษรที่ 6 แถว
  ที่ 2
  lcd.print("My ESP32");
}
void loop() {
```




13. การใช้งาน Relay

ตัวอย่างโค้ดที่ 15 การใช้งานโมดูล Relay เปิด-ปิด ไฟ

```
int led = 2;
void setup() {
  pinMode(led, OUTPUT);
}

void loop() {
  digitalWrite(led, HIGH);
  delay(1000);
  digitalWrite(led, LOW);
  delay(1000);
}
```




14. การใช้งาน Servo Motor

ตัวอย่างโค้ดที่ 16 การใช้งาน Servo Motor แบบ 0-180 องศา

Library ที่ใช้: ESP32Servo(by Kevin Harrington...) v3.0.9


```
#include <ESP32Servo.h>
Servo myservo; //ประกาศตัวแปรแทน Servo
void setup() {
  myservo.attach(4); // กำหนดขา 4 ควบคุม Servo
}
void loop() {
  myservo.write(0); // หมุนไปองศาที่ 0
  delay(1000); // หน่วงเวลา 1000ms
  myservo.write(90); // หมุนไปองศาที่ 90
  delay(1000); // หน่วงเวลา 1000ms
  myservo.write(180); // หมุนไปองศาที่ 180
  delay(1000); // หน่วงเวลา 1000ms
}
```



15. การใช้งานเซนเซอร์วัดระยะทาง

ตัวอย่างโค้ดที่ 17 การใช้งานเซนเซอร์วัดระยะทางด้วย Ultrasonic Module HC-SR04

```
int TRIG_PIN = 14;
int ECHO_PIN = 27;
float duration_us, distance_cm;
void setup() {
  Serial.begin (115200);
  pinMode(TRIG_PIN, OUTPUT);
  pinMode(ECHO_PIN, INPUT);
}
void loop() {
  digitalWrite(TRIG_PIN, HIGH);
  delayMicroseconds(10);
  digitalWrite(TRIG_PIN, LOW);
  duration_us = pulseIn(ECHO_PIN, HIGH);
  distance_cm = 0.017 * duration_us;
  Serial.print("distance: ");
  Serial.print(distance_cm);
  Serial.println(" cm");
  delay(500);
}
```




16. การใช้งานเซนเซอร์วัดอุณหภูมิและความชื้น

ตัวอย่างโค้ดที่ 18 การใช้งานเซนเซอร์วัดอุณหภูมิและความชื้น DHT11

Library ที่ใช้: 1. Adafruit_Unified_Sensor(by Adafruit) v1.1.15
2. DHT_sensor_library(by Adafruit)v1.4.6


```
#include "DHT.h"
#define DHTPIN 2
#define DHTTYPE DHT11
DHT dht(DHTPIN, DHTTYPE);
void setup() {
  Serial.begin(115200);
  dht.begin();
}
void loop() {
  delay(2000);
  float h = dht.readHumidity();
  float t = dht.readTemperature();
  float f = dht.readTemperature(true);
  if (isnan(h) || isnan(t) || isnan(f)) {
    Serial.println("Failed to read from DHT sensor!");
    return;
  }
  float hic = dht.computeHeatIndex(t, h, false);
  float hif = dht.computeHeatIndex(f, h);
  Serial.println("Humidity: " + String(h) + "% ");
  Serial.println("Temperature: " + String(t) + " C / " + String(f) + " F");
  Serial.println("Heat index: " + String(hic) + " C / " + String(hif) + " F");
  Serial.println("-----");
}
```



17. การใช้งานเซนเซอร์วัดความชื้นในดิน

ตัวอย่างโค้ดที่ 19 การใช้งานเซนเซอร์วัดความชื้นในดิน Soil Moisture Sensor


```
int sensorPin = 2;
void setup() {
  pinMode(sensorPin, INPUT);
  Serial.begin(115200);
}
void loop() {
  int val = analogRead(sensorPin);
  Serial.println("val = " + String(val));
  delay(100);
}
```



18. การใช้งานเปิดปิดปั้มน้ำผ่าน Relay

ตัวอย่างโค้ดที่ 20 การเปิดปิดปั้มน้ำผ่าน Relay


```
int relay = 14;
void setup() {
  pinMode(relay, OUTPUT);
}
void loop() {
  digitalWrite(relay, HIGH);
  delay(1000);
  digitalWrite(relay, LOW);
  delay(1000);
}
```



19. การใช้งานควบคุม Motor ผ่านโมดูลขับ L298N

ตัวอย่างโค้ดที่ 21 การควบคุม Motor ด้วยโมดูลขับ L298N ผ่านคำสั่ง Serial

```
int ml_pin = 27, mr_pin = 14, speed_pin = 12;
int ml = 0, mr = 0, speed = 0;
void setup() {
  pinMode(ml_pin, OUTPUT);
  pinMode(mr_pin, OUTPUT);
  pinMode(speed_pin, OUTPUT);
  Serial.begin(115200);
}
void loop() {
  if (Serial.available()) {
    String data = Serial.readString(); data.trim();
    Serial.print("Received: "); Serial.println(data);
    if( data=="left" ) {
      ml = 1; mr = 0;
    } else if( data=="right" ) {
      ml = 0; mr = 1;
    } else if(data=="stop") {
      ml = 0; mr = 0;
    } else {
      int x = data.toInt(); Serial.println(x);
      speed = x;
    }
  }
  Serial.println(String(ml) + ", " + String(mr) + ", " + String(speed));
  digitalWrite(ml_pin, ml);
  digitalWrite(mr_pin, mr);
  analogWrite(speed_pin, speed);
  delay(50);
}
```



พิมพ์คำสั่งผ่าน Serial

left : เพื่อหมุนตามเข็มนาฬิกา

right : เพื่อหมุนทวนเข็มนาฬิกา

stop: เพื่อหยุด

เลขจำนวนเต็ม 0-255 : เพื่อปรับความเร็ว

20. การใช้งานเซนเซอร์ RFID


ตัวอย่างโค้ดที่ 22 การใช้งานเซนเซอร์ RFID

Library ที่ใช้: MFRC522v2 (by Github Community) v2.0.6

```
#include <MFRC522v2.h>
#include <MFRC522DriverSPI.h>
#include <MFRC522DriverPinSimple.h>
MFRC522DriverPinSimple ss_pin(5);
MFRC522DriverSPI driver{ss_pin};
MFRC522 mfr522{driver};

void setup() {
  Serial.begin(115200);
  mfr522.PCD_Init();
}

void loop() {
  if( mfr522.PICC_IsNewCardPresent() && mfr522.PICC_ReadCardSerial() ) {
    String uid = "";
    for (byte i = 0; i < mfr522.uid.size; i++) {
      if (mfr522.uid.uidByte[i] < 0x10) {
        uid += "0";
      }
      uid += String(mfr522.uid.uidByte[i], HEX);
    }
    Serial.println(uid);
  }
}
```




21. การประยุกต์ใช้ระบบรดน้ำต้นไม้อัตโนมัติ

ตัวอย่างโค้ดที่ 23 การประยุกต์ใช้ระบบรดน้ำต้นไม้อัตโนมัติ โดยใช้เซนเซอร์ตรวจจับความชื้นในดิน

```
int sensorPin = 2;
int relay = 14;
void setup() {
  Serial.begin(115200);
  pinMode(sensorPin, INPUT);
  pinMode(relay, OUTPUT);
}
void loop() {
  int val = analogRead(sensorPin);
  Serial.println("val = " + String(val));
  if( val<200 ) {
    digitalWrite(relay, HIGH);
  } else {
    digitalWrite(relay, LOW);
  }
  delay(100);
}
```

วงจรไฟฟ้า




22. การประยุกต์ใช้ระบบอ่างล้างมืออัตโนมัติ

ตัวอย่างโค้ดที่ 24 การประยุกต์ใช้ระบบอ่างล้างมือ โดยใช้เซนเซอร์ตรวจจับระยะด้วย Ultrasonic

```
int TRIG_PIN = 4;
int ECHO_PIN = 2;
int RELAY_PIN = 14;
void setup() {
  Serial.begin (115200);
  pinMode(TRIG_PIN, OUTPUT);
  pinMode(ECHO_PIN, INPUT);
  pinMode(RELAY_PIN, OUTPUT);
}
void loop() {
  digitalWrite(TRIG_PIN, HIGH);
  delayMicroseconds(10);
  digitalWrite(TRIG_PIN, LOW);
  float duration_us = pulseIn(ECHO_PIN, HIGH);
  float distance_cm = 0.017 * duration_us;
  Serial.print("distance: ");
  Serial.print(distance_cm);
  Serial.println(" cm");
  if( distance_cm < 20 ) {
    digitalWrite(RELAY_PIN, HIGH);
  } else {
    digitalWrite(RELAY_PIN, LOW);
  }
  delay(500);
}
```

วงจรไฟฟ้า




23. การประยุกต์ใช้ระบบวัดอุณหภูมิและความชื้นพร้อมแสดงบนจอ LCD และไฟสถานะ

ตัวอย่างโค้ดที่ 25 การประยุกต์ใช้ระบบวัดอุณหภูมิและความชื้นพร้อมแสดงบนจอ LCD และไฟสถานะ

```
#include <LiquidCrystal_I2C.h>
LiquidCrystal_I2C lcd(0x27, 16, 2);
#include "DHT.h"
#define DHTPIN 2
#define DHTTYPE DHT11
int LED_R_PIN = 27;
int LED_G_PIN = 14;
DHT dht(DHTPIN, DHTTYPE);
void setup() {
  Serial.begin(115200);
  lcd.init();
  lcd.backlight();
  lcd.setCursor(0, 0);    // ไปที่ตัวอักษรที่ 0 แถวที่ 1
  lcd.print("Welcome");
  delay(1000);
  dht.begin();
  pinMode(LED_R_PIN, OUTPUT);
  pinMode(LED_G_PIN, OUTPUT);
  lcd.clear();
}
void loop() {
  delay(2000);
  float h = dht.readHumidity();
  float t = dht.readTemperature();
  float f = dht.readTemperature(true);
  if (isnan(h) || isnan(t) || isnan(f)) {
    Serial.println("Failed to read from DHT sensor!");
    return;
  }
  float hic = dht.computeHeatIndex(t, h, false);
  float hif = dht.computeHeatIndex(f, h);
  Serial.println("Humidity: " + String(h) + "% ");
  Serial.println("Temperature: " + String(t) + " C / " + String(f) + " F");
  Serial.println("Heat index: " + String(hic) + " C / " + String(hif) + " F");
  Serial.println("-----");
  lcd.setCursor(0, 0);    // ไปที่ตัวอักษรที่ 0 แถวที่ 1
  lcd.print("Humi : " + String(h) + " % ");
  lcd.setCursor(0, 1);    // ไปที่ตัวอักษรที่ 0 แถวที่ 2
  lcd.print("Temp : " + String(t) + " C ");
  if( h>35 ) {
    digitalWrite(LED_R_PIN, HIGH);
    digitalWrite(LED_G_PIN, LOW);
  } else {
    digitalWrite(LED_R_PIN, LOW);
    digitalWrite(LED_G_PIN, HIGH);
  }
}
```

วงจรไฟฟ้า



24. การใช้ประยุกต์เปิด-ปิด ไฟ ผ่านมือถือ

ตัวอย่างโค้ดที่ 26 การประยุกต์เปิด-ปิด ไฟ ผ่านมือถือ

Library ที่ใช้: 1.ESPAsyncWebServer(by lacamera)v3.1.0
2.AsyncTCP(by dvarrel)v1.1.4

□ ไฟล์โปรแกรมหลัก (esp32.ino)

```
#include <WiFi.h>
#include <ESPAsyncWebServer.h>
char* ssid = "My ESP32";
char* password = "12345678";

AsyncWebServer server(80);
#include "index.h"

int RELAY_PIN = 2;

void setup() {
  Serial.begin(9600);
  pinMode(RELAY_PIN, OUTPUT);

  WiFi.softAP(ssid, password);
  delay(100);

  server.on("/", HTTP_GET, [](AsyncWebServerRequest* request) {
    String html = index_page;
    request->send(200, "text/html", html);
  });

  server.on("/on", HTTP_GET, [](AsyncWebServerRequest* request) {
    digitalWrite(RELAY_PIN, HIGH);
    request->send(200, "text/html", "OK Relay ON");
  });

  server.on("/off", HTTP_GET, [](AsyncWebServerRequest* request) {
    digitalWrite(RELAY_PIN, LOW);
    request->send(200, "text/html", "OK Relay OFF");
  });


  server.begin();
}

void loop() {
}
```

□ ไฟล์หน้าเว็บโค้ด HTML (index.h)

```
const char index_page[] PROGMEM = R"=====(
<!DOCTYPE html>
<html>
<head>
  <meta charset="utf-8">
  <meta name="viewport" content="width=device-width, initial-scale=1">
  <title>ESP32 Web Server</title>
  <style>
    body {
      font-family: sans-serif; margin: 0; padding: 50px 16px;
      text-align: center; color: #333;
    }
    h1 { margin: 0 0 16px 0; font-size: 30px; }
    h4 {
      margin: 0 0 30px 0;
      font-size: 20px;
      font-weight: normal;
      color: #555;
    }
    button {
      font-size: 30px;
      padding: 20px 24px;
      margin: 8px;
      border: none;
      border-radius: 8px;
      cursor: pointer;
      width: 140px;
      color: #fff;
    }
    button:hover { opacity: 0.9; }
    #btn-on { background: #4caf50; }
    #btn-off { background: #f44336; }
  </style>
</head>
<body>
  <h1>ยินดีต้อนรับสู่</h1>
  <h4>โครงการควบคุมเปิด-ปิดไฟ ผ่านมือถือ</h4>
  <button id="btn-on" onclick="send('/on') ">เปิดไฟ</button>
  <button id="btn-off" onclick="send('/off') ">ปิดไฟ</button>
  <script>
    function send(url) {
      var xhr = new XMLHttpRequest();
      xhr.open("GET", url, true);
      xhr.send();
    }
  </script>
</body>
</html>
)=====";
```

□ วงจรไฟฟ้า



อัปโหลดโค้ดโปรแกรม และใช้มือถือค้นหาสัญญาณ WiFi ชื่อ “My ESP32” และเชื่อมต่อด้วยรหัสผ่าน 12345678 เมื่อเชื่อมต่อสำเร็จเข้าไปเปิด Browser พิมพ์ URL : 192.168.4.1 จะปรากฏหน้าเว็บที่เราเขียนขึ้นมา และสามารถกดปุ่มเพื่อควบคุมการเปิด-ปิด ไฟได้